This maple file illustrates the use of the main procedures of the package *RationalFirstIntegrals* writen by *A. Bostan, G. Chèze, T. Cluzeau, and J.-A. Weil.*
For explanations and details about the algorithms behind the procedures, see our related paper:

Efficient algorithms for computing rational first integrals of planar polynomial vector fields.

```
> restart:
```

We load our package and the *gfun* package needed for the procedure *HeuristicRationalFirstIntegral*:

```
> with(gfun);
  with(RationalFirstIntegrals);
```

[*Laplace*, *NumGfun*, *Parameters*, *algebraicsubs*, *algeqtodiffeq*, *algeqtoseries*, *algfuntoalgeq*, *borel*, *cauchyproduct*, *diffeq*diffeq*, *diffeq+diffeq*, *diffeqtohomdiffeq*, *diffeqtorec*, *firstnonzero*, *getname*, *gfun_pade*, *guesseqn*, *guessgf*, *hadamardproduct*, *holexprtodiffeq*, *invborel*, *listprimpart*, *listtoalgeq*, *listtodiffeq*, *listtohypergeom*, *listtolist*, *listtoratpoly*, *listtorec*, *listtoseries*, *makediffeq*, *maxdegcoeff*, *maxdegeqn*, *maxindex*, *mindegcoeff*, *mindegeqn*, *minindex*, *myisolve*, *nth_term*, *pade2*, *poltodiffeq*, *poltorec*, *ratpolytocoeff*, *rec*rec*, *rec+rec*, *rectodiffeq*, *rectohomrec*, *rectoproc*, *seriestoalgeq*, *seriestodiffeq*, *seriestohypergeom*, *seriestolist*, *seriestoratpoly*, *seriestorec*, *seriestoseries*, *systomatrix*]

[*DarbouxPolynomials*, *DeterministicRationalFirstIntegral*, *GenericRationalFirstIntegral*, *GuessMinimalPolynomial*, *HeuristicRationalFirstIntegral*, *PadeHermiteHeuristic*, *ProbabilisticRationalFirstIntegral*, *SimplifyRFI*, *simpli1*, *simpli2*] **(1)**

We define the planar polynomial vector field {x'=A(x,y), y'=B(x,y)} by giving the polynomials A and B:

```
> A := 4*(x-1)*(x+1) ;
> B := 1+(-4*x^2+4)*y^2-4*x*y ;
```

$$A := 4\,(x-1)\,(x+1)$$
$$B := 1+\left(-4\,x^2+4\right)y^2-4\,x\,y \qquad (2)$$

We need a point *pt* for which A(pt,y) <> 0:

```
> pt:=0;
  check_pt:=eval(A,x=pt);
```

$$pt := 0$$
$$check\_pt := -4 \qquad (3)$$

COMPUTATION OF RATIONAL FIRST INTEGRALS

We can now use our procedures to search for rational first integrals of degree less than or equal to a given *N:*

```
> N:=4;
```

$$N := 4 \quad \textbf{(4)}$$

The first procedure is called *GenericRationalFirstIntegral*:

```
> GenericRationalFirstIntegral(A,B,pt,N);
```

"None" **(5)**

This means that the planar polynomial vector field {x'=A(x,y), y'=B(x,y)} does not have any rational first integral of degree
less than or equal to 4 but we can of course search for a rational first integral of greater degree:

```
> N:=5;
```

$$N := 5 \quad \textbf{(6)}$$

```
> GenericRationalFirstIntegral(A,B,pt,N);
  F:=%:
```

$$\frac{1}{4}\,\frac{4\,x^3\,y^2-4\,x^2\,y-4\,x\,y^2+x+4\,y}{4\,x^3\,y^2+4\,x^2\,y^2-4\,x^2\,y-4\,x\,y^2-4\,y^2+x+4\,y-1} \quad \textbf{(7)}$$

Here, the planar polynomial vector field {x'=A(x,y), y'=B(x,y)} has a rational first integral of degree N=5. We can check this:

```
> check_RFI:=simplify(A*diff(F,x)+B*diff(F,y));
```

$$check\_RFI := 0 \quad \textbf{(8)}$$

In practice, this first procedure is not very efficient so it is better to use one of the next procedures:

For example, one may use our probabilistic algorithm *ProbabilisticRationalFirstIntegral.*
This procedure requires two rational numbers *c1* and *c2* that can be chosen arbitrarily (so that A (pt,ci) <> 0):

```
> c1:=0;
  c2:=1;
```

$$c1 := 0$$

$$c2 := 1 \quad \textbf{(9)}$$

```
> N:=4;
  ProbabilisticRationalFirstIntegral(A,B,pt,[c1,c2],N);
```

$$N := 4$$

"None" **(10)**

```
> N:=5;
  ProbabilisticRationalFirstIntegral(A,B,pt,[c1,c2],N);F:=%:
  check_RFI:=simplify(A*diff(F,x)+B*diff(F,y));
```

$$N := 5$$

$$-\frac{2\,(12\,x^3\,y^2+8\,x^2\,y^2-12\,x^2\,y-12\,x\,y^2-8\,y^2+3\,x+12\,y-2)}{20\,x^3\,y^2+16\,x^2\,y^2-20\,x^2\,y-20\,x\,y^2-16\,y^2+5\,x+20\,y-4}$$

$$check\_RFI := 0 \quad \textbf{(11)}$$

One may also use our deterministic algorithm *DeterministicRationalFirstIntegral* which runs recursively *ProbabilisticRationalFirstIntegral.*
This procedure requires an interval [a,b] in which the rational numbers needed for *ProbabilisticRationalFirstIntegral* are chosen.

```
> a:=-1;
  b:=1;
```

$$a := -1$$

$$b := 1 \quad \textbf{(12)}$$

```
> N:=4;
  DeterministicRationalFirstIntegral(A,B,pt,[a,b],N);
```

$$N := 4$$

$$\text{"None"} \tag{13}$$

```
> N:=5;
  DeterministicRationalFirstIntegral(A,B,pt,[a,b],N);F:=%:
  check_RFI:=simplify(A*diff(F,x)+B*diff(F,y));
```

$$N := 5$$

$$-\frac{4\left(4x^3y^2-4x^2y^2-4x^2y-4xy^2+4y^2+x+4y+1\right)}{20x^3y^2-16x^2y^2-20x^2y-20xy^2+16y^2+5x+20y+4}$$

$$check\_RFI := 0 \tag{14}$$

Finally, one may use our heuristic *HeuristicRationalFirstIntegral*. This procedure requires the *gfun* package.
This procedure requires two rational numbers *c1* and *c2* that can be chosen arbitrarily (so that A (pt,ci) <> 0):

```
> c1:=-1;
  c2:=1;
```

$$c1 := -1$$

$$c2 := 1 \tag{15}$$

```
> N:=3;
  HeuristicRationalFirstIntegral(A,B,pt,[c1,c2],N);
```

$$N := 3$$

$$\text{"Pade-Hermite fails!"} \tag{16}$$

Here our heuristic fails so that we can not conclude.

```
> N:=4;
  HeuristicRationalFirstIntegral(A,B,pt,[c1,c2],N);
  F:=%:
  check_RFI:=simplify(A*diff(F,x)+B*diff(F,y));
```

$$N := 4$$

$$\left(4\left(4c1^2x^3y^2-4c2^2x^3y^2-4c1^2x^2y-4c1^2xy^2+4c1x^2y^2+4c2^2x^2y+4c2^2xy^2 -4c2x^2y^2+c1^2x+4c1^2y-4c1y^2-c2^2x-4c2^2y+4c2y^2-c1+c2\right)\right) \Big/ \left(16c2^2x^3y^2-16c2^2x^2y-16c2^2xy^2+16c2x^2y^2+4x^3y^2+4c2^2x+16c2^2y -16c2y^2-4x^2y-4xy^2-4c2+x+4y\right)$$

$$check\_RFI := 0 \tag{17}$$

Note that *gfun* is not using a notion of global degree so that, contrary to the previous procedures, we find a rational first integral by running
the procedure *HeuristicRationalFirstIntegral* with N=4 since the degree of the rational first integral in each variable x and y is lower than 4!

SIMPLIFICATION OF RATIONAL FIRST INTEGRALS

Our package includes a procedure called *SimplifyRFI* which can be used to obtain a rational first integral written in more compact (simpler) form

than the ones returned by the previous procedures.

```
> F1:=SimplifyRFI(A,B,F);
```

$$F1 := \frac{(x+1)\,(2\,x\,y-2\,y-1)^2}{(x-1)\,(2\,x\,y+2\,y-1)^2} \quad (18)$$

```
> check_RFI:=simplify(A*diff(F1,x)+B*diff(F1,y));
```

$$check\_RFI := 0 \quad (19)$$

COMPUTATION OF DARBOUX POLYNOMIALS

Our package also inlcude a last procedure for computing all the irreducible Darboux polynomials of the planar polynomial vector field {
x'=A(x,y), y'=B(x,y)} of degree lower than or equal to a given bound N.

```
> N:=4;
  Darboux:=[DarbouxPolynomials(A,B,pt,N)];
```

$$N := 4$$

$$Darboux := \left[-\frac{1}{2}+(x-1)\,y,\ -\frac{1}{4}+\left(x^2-1\right)y^2,\ -\frac{1}{2}+(x+1)\,y\right] \quad (20)$$

Here we have a set of 3 irreducible Darboux polynomials. This can be checked by computing the corresponding cofactors.

```
> Cofactors:= [seq( normal(simplify(A*diff(Darboux[i],x)+B*diff
  (Darboux[i],y))/Darboux[i]),i=1..nops(Darboux))];
```

$$Cofactors := \left[-4\,x^2\,y-2\,x+4\,y+2,\ -8\left(x^2-1\right)y,\ -4\,x^2\,y-2\,x+4\,y-2\right] \quad (21)$$

```
> N:=5;
  Darboux:=[DarbouxPolynomials(A,B,pt,N)];
```

$$N := 5$$

$$Darboux := \left[\text{"Infinite number of Darboux polynomials"},\ \left(x^3+\frac{4\,c\,x^2}{4\,c^2+1}-x-\frac{4\,c}{4\,c^2+1}\right)y^2 + \left(-x^2+1\right)y+\frac{1}{4}\,x-\frac{c}{4\,c^2+1}\right] \quad (22)$$

As there exists a rational first integral of degree 5 (see the above calculations), we know from the theory that there exists an infinite number of Darboux
polynomials of degree lower than or equal to 5. The second output of the *DarbouxPolynomials* procedure is a "generic" Darboux polynomial indexed by c.
We can check that the cofactor does not depend on c (i.e., on the Darboux polynomial).

```
> Cofactor:=normal(simplify(A*diff(Darboux[2],x)+B*diff(Darboux[2],y)
  )/Darboux[2]);
```

$$Cofactor := -8\left(x^2-1\right)y \quad (23)$$

Note that, as *GenericRationalFirstIntegral,* the procedure *DarbouxPolynomial* is not very efficient so that it is difficult to use it when A and B have high degrees
or with a high value of N!